```
> restart;
```

Gosper's algorithm

```
> u:=(3*n+1)/(n+1)*binomial(2*n,n);
```

$$u := \frac{(3n+1)\binom{2n}{n}}{n+1} \quad (1.1)$$

```
> factor(normal(expand(subs(n=n+1,u)/u)));
```

$$\frac{2(3n+4)(2n+1)}{(3n+1)(n+2)} \quad (1.2)$$

```
> Z,A,B,C:=SumTools[Hypergeometric][PolynomialNormalForm](%,n);
```

$$Z, A, B, C := 4, n+\frac{1}{2}, n+2, n+\frac{1}{3} \quad (1.3)$$

Check

```
> normal(Z*A/B*subs(n=n+1,C)/C);
```

$$\frac{2(3n+4)(2n+1)}{(3n+1)(n+2)} \quad (1.4)$$

```
> rec:=Z*A*X(n+1)-subs(n=n-1,B)*X(n)=C;
```

$$rec := 4\left(n+\frac{1}{2}\right)X(n+1)-(n+1)X(n) = n+\frac{1}{3} \quad (1.5)$$

```
> LREtools[polysols](rec,X(n),{});
```

$$\frac{1}{3} \quad (1.6)$$

```
> R:=%*subs(n=n-1,B)/C;
```

$$R := \frac{n+1}{3\left(n+\frac{1}{3}\right)} \quad (1.7)$$

```
> sumu:=normal(R*u);
```

$$sumu := \binom{2n}{n} \quad (1.8)$$

Check:

```
> normal(expand(subs(n=n+1,sumu)-sumu));
```

$$\frac{(3n+1)\binom{2n}{n}}{n+1} \quad (1.9)$$

Zeilberger's algorithm

```
> restart;
> u:=binomial(n+k,k)^2*binomial(n,k)^2;
```

$$u := \binom{n+k}{k}^2\binom{n}{k}^2 \quad (2.1)$$

```
> telescoper:=add(a[i](n)*U(n+i,k),i=0..2);
```

$$telescoper := a_0(n)\,U(n,k)+a_1(n)\,U(n+1,k)+a_2(n)\,U(n+2,k) \quad (2.2)$$

```
> collect(expand(eval(%,U=unapply(u,[n,k])))/u,a,normal); # this
uses U(n+1,k)/U(n,k)
```

(2.3)

$$a_0(n) + \frac{(n+1+k)^2 a_1(n)}{(-n-1+k)^2} + \frac{(n+2+k)^2 (n+1+k)^2 a_2(n)}{(-n-2+k)^2 (-n-1+k)^2} \quad \textbf{(2.3)}$$

Apply Gosper's algorithm to this times U(n,k)

```
> eval(%,k=k+1)/%*normal(expand(subs(k=k+1,u))/u); # this uses U
  (n,k+1)/U(n,k)
```

$$\left(\left(a_0(n) + \frac{(n+2+k)^2 a_1(n)}{(-n+k)^2} + \frac{(n+3+k)^2 (n+2+k)^2 a_2(n)}{(-n-1+k)^2 (-n+k)^2}\right) (n+1 + k)^2 (-n+k)^2\right) \Big/ \left(\left(a_0(n) + \frac{(n+1+k)^2 a_1(n)}{(-n-1+k)^2} + \frac{(n+2+k)^2 (n+1+k)^2 a_2(n)}{(-n-2+k)^2 (-n-1+k)^2}\right) (k+1)^4\right) \quad \textbf{(2.4)}$$

```
> normal(%):
```

splits into two parts:

```
> r1,r2:=selectremove(has,%,a);
```

$$r1, r2 := \left(-4 a_1(n) k + 12 a_1(n) n + 60 a_2(n) k + 60 a_2(n) n + a_0(n) k^4 + a_0(n) n^4 + a_1(n) k^4 + a_1(n) n^4 + a_2(n) k^4 + a_2(n) n^4 - 2 a_0(n) k^3 + 2 a_0(n) n^3 + 2 a_1(n) k^3 + 6 a_1(n) n^3 + 10 a_2(n) k^3 + 10 a_2(n) n^3 + a_0(n) k^2 + a_0(n) n^2 - 3 a_1(n) k^2 + 13 a_1(n) n^2 + 37 a_2(n) k^2 + 37 a_2(n) n^2 - 4 a_0(n) k^3 n + 6 a_0(n) k^2 n^2 - 4 a_0(n) k n^3 - 2 a_1(n) k^2 n^2 + 4 a_2(n) k^3 n + 6 a_2(n) k^2 n^2 + 4 a_2(n) k n^3 + 6 a_0(n) k^2 n - 6 a_0(n) k n^2 - 6 a_1(n) k^2 n - 2 a_1(n) k n^2 + 30 a_2(n) k^2 n + 30 a_2(n) k n^2 - 2 a_0(n) k n - 6 a_1(n) k n + 74 a_2(n) k n + 4 a_1(n) + 36 a_2(n)\right) \Big/ \left(4 a_1(n) k + 12 a_1(n) n + 12 a_2(n) k + 12 a_2(n) n - 12 a_0(n) k + 12 a_0(n) n + a_0(n) k^4 + a_0(n) n^4 + a_1(n) k^4 + a_1(n) n^4 + a_2(n) k^4 + a_2(n) n^4 - 6 a_0(n) k^3 + 6 a_0(n) n^3 - 2 a_1(n) k^3 + 6 a_1(n) n^3 + 6 a_2(n) k^3 + 6 a_2(n) n^3 + 13 a_0(n) k^2 + 13 a_0(n) n^2 - 3 a_1(n) k^2 + 13 a_1(n) n^2 + 13 a_2(n) k^2 + 13 a_2(n) n^2 - 4 a_0(n) k^3 n + 6 a_0(n) k^2 n^2 - 4 a_0(n) k n^3 - 2 a_1(n) k^2 n^2 + 4 a_2(n) k^3 n + 6 a_2(n) k^2 n^2 + 4 a_2(n) k n^3 + 18 a_0(n) k^2 n - 18 a_0(n) k n^2 - 6 a_1(n) k^2 n + 2 a_1(n) k n^2 + 18 a_2(n) k^2 n + 18 a_2(n) k n^2 - 26 a_0(n) k n + 6 a_1(n) k n + 26 a_2(n) k n + 4 a_0(n) + 4 a_1(n) + 4 a_2(n)\right), \frac{(n+1+k)^2 (-n-2+k)^2}{(k+1)^4} \quad \textbf{(2.5)}$$

the first one is of the form P(k+1)/P(k):

```
> P:=collect(denom(r1),a,factor); normal(r1-subs(k=k+1,P)/P);
```

$$P := (-n-2+k)^2 (-n-1+k)^2 a_0(n) + (-n-2+k)^2 (n+1+k)^2 a_1(n) + (n$$

$$+2+k)^2\,(n+1+k)^2\,a_2(n)$$

$$0 \tag{2.6}$$

the second one can be decomposed by Gosper-Petkovšek decomposition:

```
> Z,A,B,C:=SumTools[Hypergeometric][PolynomialNormalForm](r2,k);
```

$$Z, A, B, C := 1,\ (-n-2+k)^2\,(n+1+k)^2,\ (k+1)^4,\ 1 \tag{2.7}$$

Changing R into B(k-1)/C(k)/P(k) * X(k) we are reduced to looking for polynomial solutions of

```
> rec:=Z*A*X(k+1)-subs(k=k-1,B)*X(k)=C*P;
```

$$rec := (-n-2+k)^2\,(n+1+k)^2\,X(k+1) - k^4\,X(k) = (-n-2+k)^2\,(-n-1+k)^2\,a_0(n) + (-n-2+k)^2\,(n+1+k)^2\,a_1(n) + (n+2+k)^2\,(n+1+k)^2\,a_2(n) \tag{2.8}$$

Bound on the degree of polynomial solutions in k:

```
> MultiSeries:-asympt(expand(eval(op(1,rec),X=unapply(k^p,k))
  /k^p),k,2);
```

$$(p-2)\,k^3 + \mathrm{O}\left(k^2\right) \tag{2.9}$$

has to be of degree 2. Undeterminate coefficients:

```
> eval(op(1,rec)-op(2,rec),X=unapply(add(x[i]*k^i,i=0..2),k));
```

$$(-n-2+k)^2\,(n+1+k)^2\left((k+1)^2\,x_2 + (k+1)\,x_1 + x_0\right) - k^4\left(x_2\,k^2 + x_1\,k + x_0\right) - (-n-2+k)^2\,(-n-1+k)^2\,a_0(n) - (-n-2+k)^2\,(n+1+k)^2\,a_1(n) - (n+2+k)^2\,(n+1+k)^2\,a_2(n) \tag{2.10}$$

lead to a linear system by taking coefficients of powers of k:

```
> sys:={coeffs(expand(%),k)};
```

$$sys := \{-x_1 - 6\,x_2 - 2\,n^2\,x_2 - 6\,n\,x_2 - a_0(n) - a_1(n) - a_2(n),\ -2\,x_0 - 5\,x_1 - 4\,x_2 - 2\,n^2\,x_1 - 6\,n\,x_2 - 6\,n\,x_1 - 2\,n^2\,x_2 + 6\,a_0(n) + 2\,a_1(n) - 6\,a_2(n) + 4\,a_0(n)\,n - 4\,a_2(n)\,n,\ -3\,x_0 + x_1 + 9\,x_2 - 6\,n\,x_0 + 15\,n^2\,x_2 + 18\,n\,x_2 + 6\,n^3\,x_2 + n^4\,x_2 - 2\,n^2\,x_0 - 13\,a_0(n) + 3\,a_1(n) - 13\,a_2(n) - 6\,a_0(n)\,n^2 + 2\,a_1(n)\,n^2 - 6\,a_2(n)\,n^2 - 18\,a_0(n)\,n + 6\,a_1(n)\,n - 18\,a_2(n)\,n,\ 4\,x_0 + 12\,x_2 + 8\,x_1 + 2\,n^4\,x_2 + 18\,n\,x_1 + 12\,n^3\,x_2 + 28\,n^2\,x_2 + 30\,n\,x_2 + 6\,n\,x_0 + 6\,n^3\,x_1 + 2\,n^2\,x_0 + 15\,n^2\,x_1 + n^4\,x_1 - 4\,a_1(n) - 12\,a_2(n) + 12\,a_0(n) + 4\,a_0(n)\,n^3 - 4\,a_2(n)\,n^3 + 18\,a_0(n)\,n^2 - 2\,a_1(n)\,n^2 - 18\,a_2(n)\,n^2 + 26\,a_0(n)\,n - 6\,a_1(n)\,n - 26\,a_2(n)\,n,\ n^4\,x_0 + n^4\,x_1 + n^4\,x_2 + 6\,n^3\,x_0 + 6\,n^3\,x_1 + 6\,n^3\,x_2 + 13\,n^2\,x_0 + 13\,n^2\,x_1 + 13\,n^2\,x_2 + 12\,n\,x_0 + 12\,n\,x_1 + 12\,n\,x_2 + 4\,x_0 + 4\,x_1 + 4\,x_2 - 12\,a_1(n)\,n - 12\,a_2(n)\,n - 12\,a_0(n)\,n - a_0(n)\,n^4 - a_1(n)\,n^4 - a_2(n)\,n^4 - 6\,a_0(n)\,n^3 - 6\,a_1(n)\,n^3 - 6\,a_2(n)\,n^3 - 13\,a_0(n)\,n^2 - 13\,a_1(n)\,n^2 - 13\,a_2(n)\,n^2 - 4\,a_0(n) - 4\,a_1(n) - 4\,a_2(n)\} \tag{2.11}$$

Solve linear system

```
> sol:=solve(sys,{seq(x[i],i=0..2),seq(a[i](n),i=0..2)});
```

$$sol := \left\{x_0 = -2\left(n^2+3n+2\right)x_2, x_1 = -\frac{3x_2}{2}, x_2 = x_2, a_0(n) = \frac{\left(n^3+3n^2+3n+1\right)x_2}{8(3+2n)}, a_1(n) = -\frac{x_2\left(17n^2+51n+39\right)}{8}, a_2(n) = \frac{x_2\left(n^3+6n^2+12n+8\right)}{8(3+2n)}\right\} \tag{2.12}$$

Thus, we have found

```
> tel:=subs(sol,x[2]=1,telescoper);
```

$$tel := \frac{\left(n^3+3n^2+3n+1\right)U(n,k)}{8(3+2n)} - \frac{\left(17n^2+51n+39\right)U(n+1,k)}{8} + \frac{\left(n^3+6n^2+12n+8\right)U(n+2,k)}{8(3+2n)} \tag{2.13}$$

```
> cert:=subs(sol,x[2]=1,subs(k=k-1,B)/C*add(x[i]*k^i,i=0..2)*
  tel/P);
```

$$cert := \left(k^4\left(k^2-\frac{3}{2}k-2n^2-6n-4\right)\left(\frac{\left(n^3+3n^2+3n+1\right)U(n,k)}{8(3+2n)} - \frac{\left(17n^2+51n+39\right)U(n+1,k)}{8} + \frac{\left(n^3+6n^2+12n+8\right)U(n+2,k)}{8(3+2n)}\right)\right)\Bigg/\left(\frac{(-n-2+k)^2(-n-1+k)^2\left(n^3+3n^2+3n+1\right)}{8(3+2n)} - \frac{(-n-2+k)^2(n+1+k)^2\left(17n^2+51n+39\right)}{8} + \frac{(n+2+k)^2(n+1+k)^2\left(n^3+6n^2+12n+8\right)}{8(3+2n)}\right) \tag{2.14}$$

Check:

```
> normal(expand(eval(tel-(subs(k=k+1,cert)-cert),U=unapply(u,[n,
  k]))));
```

$$0 \tag{2.15}$$

Simplify certificate:

```
> factor(normal(expand(eval(cert,U=unapply(u,[n,k])))/u))*u;
```

$$\frac{k^4\left(2k^2-4n^2-3k-12n-8\right)\binom{n+k}{k}^2\binom{n}{k}^2}{2(-n-2+k)^2(-n-1+k)^2} \tag{2.16}$$

Maple's implementation:

```
> SumTools[Hypergeometric][Zeilberger](u,n,k,Sn);
```

Multiple Binomial Sums

```
sumtoCT:=proc(u,inds,x,ini)
     if not has(u,inds) or type(u,`^`) and not has(op(1,u),inds) then u
     elif type(u,specfunc(Delta)) then x[ini]^op(u)
     elif type(u,specfunc(Binomial)) then (1+x[ini])^op(1,u)/x[ini]^op(2,u)
     elif type(u,`+`) then normal(map(sumtoCT,args))
     elif type(u,`*`) then sumtoCT(op(1,u),inds,x,ini)*sumtoCT(subsop(1=1,u),inds,x,ini+1)
     elif type(u,`^`) then sumtoCT(op(1,u),inds,x,ini)*sumtoCT(op(1,u)^(op(2,u)-1),inds,x,ini+1)
     elif type(u,specfunc(Sum)) then normal(sum(sumtoCT(op(1,u),inds,x,ini),op(2,u)))
     else error "forgotten case",u
     fi
end:

sumtoGF:=proc(u,inds,t)
local x, CT:=sumtoCT(u,inds,x,0),i,indx;
     _EnvFormal:=true;
     for i to nops(inds) do if has(CT,inds[i]) then CT:=sum(CT*t[i]^inds[i],inds[i]=0..infinity) fi od;
     indx:=select(has,indets(CT,indexed),x);
     CT:=normal(CT/convert(indx,`*`));
     for i in indx do CT:=Int(CT,i) od;
     CT
```

> sumtoCT(Binomial(n,k),[k,n],x,0);

$$\frac{\left(1+x_0\right)^n}{x_0^k} \tag{3.1}$$

> sumtoCT(Binomial(n,k)*Binomial(n+k,k),[k,n],x,0);

$$\frac{\left(1+x_0\right)^n \left(1+x_1\right)^{n+k}}{x_0^k\, x_1^k} \tag{3.2}$$

> sumtoCT(Sum(Binomial(n,k)*Binomial(n+k,k),k=0..n),[k,n],x,0);

$$-\frac{\left(1+x_0\right)^n \left(1+x_1\right)^n x_0\, x_1 \left(\left(\frac{1+x_1}{x_0\, x_1}\right)^{n+1}-1\right)}{x_0\, x_1 - x_1 - 1} \tag{3.3}$$

> sumtoCT(Binomial(n,k)^2*Binomial(n+k,k)^2,[k,n],x,0);

$$\frac{\left(1+x_0\right)^n \left(1+x_1\right)^n \left(1+x_1\right)^{n+k} \left(1+x_2\right)^{n+k}}{x_0^k \left(x_1^k\right)^2 x_2^k} \tag{3.4}$$

Apéry

> sumtoCT(Sum(Binomial(n,k)^2*Binomial(n+k,k)^2,k=0..n),[k,n],x,0);

$$-\frac{\left(1+x_0\right)^n \left(\left(1+x_1\right)^n\right)^2 \left(1+x_2\right)^n x_0\, x_1^2\, x_2 \left(\left(\frac{\left(1+x_1\right)\left(1+x_2\right)}{x_0\, x_1^2\, x_2}\right)^{n+1}-1\right)}{x_0\, x_1^2\, x_2 - x_1\, x_2 - x_1 - x_2 - 1} \tag{3.5}$$

> sumtoGF(Sum(Binomial(n,k)^2*Binomial(n+k,k)^2,k=0..n),[k,n],t);

$$\int\!\!\int\!\!\int x_1 \Big/ \Big(\big(t_2 x_0 x_1^2 x_2 + t_2 x_0 x_1^2 + 2 t_2 x_0 x_1 x_2 + t_2 x_1^2 x_2 + 2 t_2 x_0 x_1 + t_2 x_0 x_2 + t_2 x_1^2 + 2 t_2 x_1 x_2 + t_2 x_0 + 2 t_2 x_1 + t_2 x_2 + t_2 - 1 \big) \big(t_2 x_0 x_1^3 x_2^2 + 2 t_2 x_0 x_1^3 x_2 + 3 t_2 x_0 x_1^2 x_2^2 + t_2 x_1^3 x_2^2 + t_2 x_0 x_1^3 + 6 t_2 x_0 x_1^2 x_2 + 3 t_2 x_0 x_1 x_2^2 + 2 t_2 x_1^3 x_2 + 3 t_2 x_1^2 x_2^2 + 3 t_2 x_0 x_1^2 + 6 t_2 x_0 x_1 x_2 + t_2 x_0 x_2^2 + t_2 x_1^3 + 6 t_2 x_1^2 x_2 + 3 t_2 x_1 x_2^2 - x_0 x_1^2 x_2 + 3 t_2 x_0 x_1 + 2 t_2 x_0 x_2 + 3 t_2 x_1^2 + 6 t_2 x_1 x_2 + t_2 x_2^2 + t_2 x_0 + 3 t_2 x_1 + 2 t_2 x_2 \tag{3.6}$$

$$+ t_2)) \, dx_0 \, dx_1 \, dx_2$$

Andrews-Paule

```
> sumtoCT(Sum(Sum(Binomial(i+j,i)^2*Binomial(4*n-2*i-2*j,2*n-2*
  i),j=0..n),i=0..n),[i,j,n],x,0);
```

$$-\frac{(1+x_1)^{4n}(1+x_1)^2\left(\left(\frac{1+x_0}{1+x_1}\right)^{n+1}-1\right)x_0\left(\left(\frac{(1+x_0)x_1}{(1+x_1)x_0}\right)^{n+1}-1\right)}{x_1^{2n}(x_0-x_1)^2} \tag{3.7}$$

```
> sumtoGF(Sum(Sum(Binomial(i+j,i)^2*Binomial(4*n-2*i-2*j,2*n-2*
  i),j=0..n),i=0..n),[n,i,j],t);
```

$$\iint -\Big(x_1^2 \big(t_1^2 x_0^2 x_1^6 + 6 t_1^2 x_0^2 x_1^5 + 2 t_1^2 x_0 x_1^6 + 15 t_1^2 x_0^2 x_1^4 + 12 t_1^2 x_0 x_1^5 + t_1^2 x_1^6 + 20 t_1^2 x_0^2 x_1^3 + 30 t_1^2 x_0 x_1^4 + 6 t_1^2 x_1^5 + 15 t_1^2 x_0^2 x_1^2 + 40 t_1^2 x_0 x_1^3 + 15 t_1^2 x_1^4 + 6 t_1^2 x_0^2 x_1 + 30 t_1^2 x_0 x_1^2 + 20 t_1^2 x_1^3 + t_1^2 x_0^2 + 12 t_1^2 x_0 x_1 + 15 t_1^2 x_1^2 - x_0 x_1^3 + 2 t_1^2 x_0 + 6 t_1^2 x_1 + t_1^2\big)\Big) \Big/ \Big(\big(t_1 x_1^4 + 4 t_1 x_1^3 + 6 t_1 x_1^2 + 4 t_1 x_1 - x_1^2 + t_1\big)\big(t_1 x_0 x_1^3 + 3 t_1 x_0 x_1^2 + t_1 x_1^3 + 3 t_1 x_0 x_1 + 3 t_1 x_1^2 + t_1 x_0 + 3 t_1 x_1 - x_0 x_1 + t_1\big)\big(t_1 x_0 x_1^3 + 3 t_1 x_0 x_1^2 + t_1 x_1^3 + 3 t_1 x_0 x_1 + 3 t_1 x_1^2 + t_1 x_0 + 3 t_1 x_1 - x_1^2 + t_1\big)\big(t_1 x_0^2 x_1^2 + 2 t_1 x_0^2 x_1 + 2 t_1 x_0 x_1^2 + t_1 x_0^2 + 4 t_1 x_0 x_1 + t_1 x_1^2 + 2 t_1 x_0 + 2 t_1 x_1 - x_0 x_1 + t_1\big)\Big) \, dx_0 \, dx_1 \tag{3.8}$$